
1 Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.2.3.16.1 which is part of the 5G ENDC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Shaun Harry


Shaun.harry@keysight.com
1.1 Verification Test Summary 
Test Case: 
8.2.3.16.1
ATS Version:
iwd-TTCN3-B2019-06_D20wk25
System Simulator used:
Keysight S8704A Protocol Conformance Toolset
UE used:
Samsung Hubble S5123 and MediaTek MT6297
Verification Status:
PASS
2 Corrections required

2.1 Change 1
	Function name
	fl_TC_8_2_3_16_1_ENDC_EUTRA_TestBody

	Reason for change
	At step 1 we only need to establish SRB3, reconfiguring existing SRBs to use NR PDCP is not required 

	Summary of change
	Receive SCConfig and RBConfig for SRB3 configuration from NR PTC and send to UE.

	TTCN module
	RRCMeasurement_ENDC_EUTRA.ttcn

	MCC160 Comment
	


Before Change

	  function fl_TC_8_2_3_16_1_ENDC_EUTRA_TestBody() runs on EUTRA_5GS_PTC

  {

    // @siclog "Steps 1-2" siclog@

    f_EUTRA38_ReconfigureSRBs_NR_PDCP(eutra_Cell1, true, -);
    // @siclog "Step 3" siclog@

    // The SS transmits an RRCReconfiguration to setup periodical intra-NR measurements and reporting on NR Cell 1 via SRB3

    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger);

    // Test body now finished

    f_EUTRA_NR_WaitForCoOrd_Trigger(NR);

  }//End of fl_TC_8_2_3_16_1_TestBody()


After Change

	   function fl_TC_8_2_3_16_1_ENDC_EUTRA_TestBody() runs on EUTRA_5GS_PTC

  {

    var template (omit) octetstring v_NR_SCGConfig := omit;

    var template (omit) octetstring v_NR_RBConfig := omit;

    var EUTRA38_SecurityParamsExtd_Type v_Security_Params;
    // @siclog "Steps 1-2" siclog@

    v_NR_SCGConfig := f_EUTRA_NR_WaitForCoOrd_Data(NR, "Reconfig");

    v_NR_RBConfig := f_EUTRA_NR_WaitForCoOrd_Data(NR, "RB");

    f_EUTRA38_508_ENDC_SendReconfiguration(eutra_Cell1,

                                           -, -,

                                           v_NR_SCGConfig,

                                           v_NR_RBConfig);

    v_Security_Params := f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys(NR);

    f_EUTRA38_Security_Set(v_Security_Params);
    // @siclog "Step 3" siclog@

    // The SS transmits an RRCReconfiguration to setup periodical intra-NR measurements and reporting on NR Cell 1 via SRB3

    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger);

    // Test body now finished

    f_EUTRA_NR_WaitForCoOrd_Trigger(NR);

  }//End of fl_TC_8_2_3_16_1_TestBody()


2.2 Change 2
	Function name
	fl_TC_8_2_3_16_1_ENDC_NR_TestBody

	Reason for change
	1. Ciphering and integrity algorithm not reauired to be sent to EUTRA PTC
2. SS configuration for SRB3 should also configure NR RLC and MAC as well as PDCP
3.  RRCReconfigurationComplete should be received from EUTRA.

4. At step 3, RRCReconfiguration for Measconfig should be configured as per 38.523-1 table 8.2.3.16.1.3.3-2. Current SCGConfig at step 3 is for removing the SRB3 bearer

5. As per test condition, UE has to send 10 periodical measurement reports in 30sec.
6. Additional periodic reports received after step 5 should not fail the test case 

	Summary of change
	1. Remove sending of ciphering and integrity a;gorithm information
2. Configuring RLC and MAC along with PDCP for SRB3. 
3. Allowing NR PTC to receive RRCREconfigurationComplete from EUTRA.

4. Configuring SCGConfig at step 3 for measconfig.

5. Using f_NR_ReceivePeriodicalMeasurementReports to receive 10 periodical MRs in 30sec

6. Adding new default altstep to handle further measurement reports after step 5.

	TTCN module
	RRCMeasurement_ENDC_NR.ttcn

	MCC160 Comment
	


Before Change

	  function fl_TC_8_2_3_16_1_ENDC_NR_TestBody() runs on ENDC_NR_PTC

  {

    var template (value) CellGroupConfig v_SCGConfig;

    var template (value) RadioBearerConfig v_RadioBearerConfig1;

    var template (omit) octetstring v_EncodedSCG_Config := omit;

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerToAddModList;

    var template (value) NR_RadioBearerList_Type v_RadioBearerList;

    var PhysCellId v_PhysCellIdServing;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (value) MeasIdToAddModList v_MeasIdConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    // @siclog "Step 1" siclog@

    f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertCiphAlg2Bitstring(f_NR_AS_CipheringAlgorithm_Get())), "NR Ciphering Alg"));

    f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertIntAlg2Bitstring(f_NR_AS_IntegrityAlgorithm_Get())), "NR Integrity Alg"));
    v_RadioBearerConfig1 := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB3)}, -, -, cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));

    v_RLC_BearerToAddModList := {cs_38508_RLC_BearerConfig_SRB(tsc_SRB3)};

    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, v_RLC_BearerToAddModList);

    //SS : configure SRB1 with RLCAckProhibit

    v_RadioBearerList := {cs_NR_SS_RadioBearer_Config(tsc_NR_RbId_SRB3, cs_NR_PDCP_Configuration_RBTerminating(cs_NR_PDCP_RbConfig_ParamsSRB))};
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, v_RadioBearerList, cs_TimingInfo_Now); //Configure SRB3 on NR

    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig1, v_SCGConfig, -, -, false);  // Now send octetstrings to EUTRA
    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);

    // @siclog "Step 3" siclog@

    // The SS transmits an RRCReconfiguration to setup periodical intra-NR measurements and reporting on NR Cell 1 via SRB3

    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, { f_NR_LogicalChannelId_SRB(tsc_SRB3) });
    v_EncodedSCG_Config := bit2oct(encvalue(v_SCGConfig));

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1)};

    v_ReportConfigList := cs_NR_ReportConfigToAddModList_1Entry(tsc_NR_IdReportConfigId1, cs_38508_ReportConfigNR_Periodical);

    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};

    v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);

    SRB.send(cas_NR_SRB3_RrcPdu_REQ(nr_Cell1, -, cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, -, v_EncodedSCG_Config, v_MeasConfig)));

    // @siclog "Step 4" siclog@

    // Check: Does the UE transmits an RRCReconfigurationComplete message via SRB3?

    SRB.receive(car_NR_SRB3_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // @siclog "Step 5" siclog@

    // Check: Does the UE transmits at least 10 MeasurementReport messges to periodically report the measured RSRP value for NR Cell 1

    // via SRB3 within the next 30 seconds?

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellIdServing, tsc_NR_MeasId1, 63, 48.0, omit, false);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    // Test body now finished

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

  } // End of fl_TC_8_2_3_15_1_TestBody


After Change

	   function fl_TC_8_2_3_16_1_ENDC_NR_TestBody() runs on ENDC_NR_PTC

  {

    var template (value) CellGroupConfig v_SCGConfig;

    var template (value) RadioBearerConfig v_RadioBearerConfig1;

    var template (omit) octetstring v_EncodedSCG_Config := omit;

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerToAddModList;

    var template (value) NR_RadioBearerList_Type v_RadioBearerList;

    var PhysCellId v_PhysCellIdServing;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (value) MeasIdToAddModList v_MeasIdConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    var template (present) MeasResults v_MeasResults;
    // @siclog "Step 1" siclog@

    //REMOVED f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertCiphAlg2Bitstring(f_NR_AS_CipheringAlgorithm_Get())), "NR Ciphering Alg"));

    //REMOVED f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertIntAlg2Bitstring(f_NR_AS_IntegrityAlgorithm_Get())), "NR Integrity Alg"));
    v_RadioBearerConfig1 := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB3)}, -, -, cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));

    v_RLC_BearerToAddModList := {cs_38508_RLC_BearerConfig_SRB(tsc_SRB3)};

    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, v_RLC_BearerToAddModList);

    //SS : configure SRB1 with RLCAckProhibit

    v_RadioBearerList := {cs_NR_SS_SRB3_Config};
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, v_RadioBearerList, cs_TimingInfo_Now); //Configure SRB3 on NR

    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig1, v_SCGConfig);  // Now send octetstrings to EUTRA

    f_ENDC_GetKeysActivateSecurity(nr_Cell1, EUTRA);
    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);

    // @siclog "Step 3" siclog@

    // The SS transmits an RRCReconfiguration to setup periodical intra-NR measurements and reporting on NR Cell 1 via SRB3

    v_SCGConfig := cs_NR_CellGroupConfigMeas(tsc_NR_CellGroupId_SCG,  //@sic R5s200612 sic@

                                                   cs_NR_SpCellConfigDef(tsc_NR_CellGroupId_SCG,

                                                                       -,

                                                                       -,

                                                                       cs_NR_ServingCellConfigMeas(tsc_NR_MeasObjectId1)));
    v_EncodedSCG_Config := bit2oct(encvalue(v_SCGConfig));

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1)};

    v_ReportConfigList := cs_NR_ReportConfigToAddModList_1Entry(tsc_NR_IdReportConfigId1, cs_38508_ReportConfigNR_Periodical);

    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};

    v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);

    SRB.send(cas_NR_SRB3_RrcPdu_REQ(nr_Cell1, -, cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, -, v_EncodedSCG_Config, v_MeasConfig)));

    // @siclog "Step 4" siclog@

    // Check: Does the UE transmits an RRCReconfigurationComplete message via SRB3?

    SRB.receive(car_NR_SRB3_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // @siclog "Step 5" siclog@

    // Check: Does the UE transmits at least 10 MeasurementReport messges to periodically report the measured RSRP value for NR Cell 1

    // via SRB3 within the next 30 seconds?

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    v_MeasResults := cr_38508_MeasResults(tsc_NR_MeasId1, v_PhysCellIdServing);

    f_NR_ReceivePeriodicalMeasurementReports(nr_Cell1, 30.0, 10, cr_NR_MeasurementReport(v_MeasResults), false);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    activate(a_NR_AddDefault_IgnorePeriodicalMeasurmentReport_SRB3(nr_Cell1, cr_NR_MeasurementReport(v_MeasResults)));
    // Test body now finished

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

  } // End of fl_TC_8_2_3_15_1_TestBody


2.3 Change 3
	Function name
	f_NR_ReceivePeriodicalMeasurementReports

	Reason for change
	Current implementation handles measurement reports on SRB1 only

	Summary of change
	Adding new parameter to control SRB1 / SRB3

	TTCN module
	NR_Measurement_Functions.ttcn

	MCC160 Comment
	


Before Change

	  function f_NR_ReceivePeriodicalMeasurementReports(NR_CellId_Type p_NR_CellId,

                                          float p_TimerValue,

                                          integer p_NumReports := 1,

                                          template (present) UL_DCCH_Message p_ExpectedMeasReport

                                          ) runs on NR_BASE_PTC

    {

        var integer v_ReportCounter := 0;

        timer t_Watchdog := p_TimerValue;

        t_Watchdog.start;

        alt

        {

        [] SRB.receive(car_NR_SRB1_RrcPdu_IND(p_NR_CellId, p_ExpectedMeasReport))
           {

                v_ReportCounter := v_ReportCounter + 1;

               if (v_ReportCounter == p_NumReports)

               {

                   t_Watchdog.stop;

                   f_NR_PreliminaryPass (__FILE__, __LINE__, "All reports have been received");

               }

               else

               {

                   repeat;

               }

           }

        []t_Watchdog.timeout

        {

            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Check failed: not all reports were received in time");

        }

      }

    };


After Change

	   function f_NR_ReceivePeriodicalMeasurementReports(NR_CellId_Type p_NR_CellId,

                                          float p_TimerValue,

                                          integer p_NumReports := 1,

                                          template (present) UL_DCCH_Message p_ExpectedMeasReport,

                                          boolean p_ReceiveOnSRB1 := true
                                          ) runs on NR_BASE_PTC

    {

        var integer v_ReportCounter := 0;

        timer t_Watchdog := p_TimerValue;

        t_Watchdog.start;

        alt

        {

        [p_ReceiveOnSRB1] SRB.receive(car_NR_SRB1_RrcPdu_IND(p_NR_CellId, p_ExpectedMeasReport))
           {

                v_ReportCounter := v_ReportCounter + 1;

               if (v_ReportCounter == p_NumReports)

               {

                   t_Watchdog.stop;

                   f_NR_PreliminaryPass (__FILE__, __LINE__, "All reports have been received");

               }

               else

               {

                   repeat;

               }

           }

        [not p_ReceiveOnSRB1] SRB.receive(car_NR_SRB3_RrcPdu_IND(p_NR_CellId, p_ExpectedMeasReport))

           {

                v_ReportCounter := v_ReportCounter + 1;

               if (v_ReportCounter == p_NumReports)

               {

                   t_Watchdog.stop;

                   f_NR_PreliminaryPass (__FILE__, __LINE__, "All reports have been received");

               }

               else

               {

                   repeat;

               }

           }
        []t_Watchdog.timeout

        {

            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Check failed: not all reports were received in time");

        }

      }

    };


2.4 Change 4
	Function name
	New altstep a_NR_AddDefault_IgnorePeriodicalMeasurmentReport_SRB3

	Reason for change
	New alternate step to ignore periodical measurement reports on SRB3

	Summary of change
	New alternate step to ignore periodical measurement reports on SRB3

	TTCN module
	NR_Measurement_Functions.ttcn

	MCC160 Comment
	


New altstep
	  altstep a_NR_AddDefault_IgnorePeriodicalMeasurmentReport_SRB3(NR_CellId_Type p_CellId,

                                                           template (present) UL_DCCH_Message p_RrcPdu

                                                           ) runs on NR_BASE_PTC

  {

    [] SRB.receive( car_NR_SRB3_RrcPdu_IND( p_CellId, p_RrcPdu ) )

      {

        repeat;

      };

  };


3 Execution Log Files 
3.1 Samsung Hubble S5123
The Samsung Hubble S5123 5G UE passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test case execution log file:

TC_8_2_3_1_16_LOG.html

In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

· PICS/PIXIT parameter file:
TC_8_2_3_1_16_PIXIT.xml

3.2 MediaTek MT6297
The MediaTek MT6297 5G UE passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test case execution log file:

TC_8_2_3_16_1_LOG.html

In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

· PICS/PIXIT parameter file:
TC_8_2_3_16_1_PIXIT.xml

4 References

